

ESTADÍSTICAS EN EXCEL

PARA INGENIERÍA Y CIENCIAS



ENFOQUE EN EJERCICIOS PRÁCTICOS

"Sin datos eres solo otra persona con una opinión."

W. Edwards Deming

VISTA PREVIA DE LOS EJERCICIOS DEL LIBRO

PRUEBA T PARA LA INFERENCIA DE LA MEDIA POBLACIONAL

=====

2.12.- Los anillos separadores para un cierto eje de transmisión tienen una especificación de espesor de 38.98-39.02 mm. La máquina que produce los anillos debe calibrarse cada cierto tiempo para asegurar que la media de los espesores esté justo en el centro de los valores mínimo y máximo permitidos. En una muestra de 6 anillos se obtuvieron los siguientes espesores:

39.030 38.997 39.012 39.008 39.019 39.002

- a).- Obtenga el estadístico de prueba t y el valor de t crítica con un nivel de confianza del 90%. Indique si estos valores muestran que la máquina requiere calibración.*
- b).- Obtenga el valor de p y el valor de p crítica con el mismo nivel de confianza que en el inciso anterior. Indique si estos valores indican que la máquina requiere calibración.*

Se trata de una prueba t para la media de una muestra. Copie los datos en una nueva hoja de Excel y obtenga los datos iniciales:

	A	B	C	D	E	F	G	H	I
1	39.03	38.997	39.012	39.008	39.019	39.002		Límite superior espec	39.02
2								Límite inferior espec	38.98
3								Nivel de confianza	0.9
4								Media propuesta	39
5								Tamaño de muestra	6
6								Media muestral	39.0113
7								Desviación estándar muestral	0.0119

Celda	Fórmula	Resultado
I4	=I2+(I1-I2)/2	39
I5	=COUNT(A:F)	6
I6	=AVERAGE(A:F)	39.0113
I7	=STDEV.S(A:F)	0.0119

Se requiere que la media poblacional sea igual a 39 y las desviaciones tanto hacia arriba como hacia abajo indicarían que la máquina requiere calibración, por lo que se trata de una prueba de dos colas.

Para el primer enfoque con valores t, se obtiene primero el estadístico de prueba t_0 y después el estadístico t crítico de la siguiente manera:

	A	B	C	D	E	F	G	H	I
1	39.03	38.997	39.012	39.008	39.019	39.002		Límite superior espec	39.02
2								Límite inferior espec	38.98
3								Nivel de confianza	0.9
4								Media propuesta	39
5								Tamaño de muestra	6
6								Media muestral	39.0113
7								Desviación estándar muestral	0.0119
8									
9								t_0	2.3275
10								t crítica	2.0150
11									RECHAZAR H0

Celda	Fórmula	Resultado
I9	=(I6-I4)/(I7/SQRT(I5))	2.3275
I10	=T.INV.2T(1-I3,I5-1)	2.0150
I11	=IF(ABS(I9)>I10,"RECHAZAR H0","NO RECHAZAR H0")	RECHAZAR H0

Observe el uso de la fórmula T.INV para la obtención del valor crítico de t . En una prueba de dos colas como esta, se rechazará la hipótesis nula solo si el valor de t_0 no está entre $\pm t_{crítica}$. En este caso 2.3275 no está entre ± 2.0150 por lo que se debe rechazar la hipótesis nula. Esto se muestra en la fórmula de la celda I11.

Para el segundo enfoque, el valor del estadístico de p se obtiene directamente del valor de t . En este caso el valor crítico de p se obtiene también directamente, del nivel de confianza deseado:

	A	B	C	D	E	F	G	H	I
1	39.03	38.997	39.012	39.008	39.019	39.002		Límite superior espec	39.02
2								Límite inferior espec	38.98
3								Nivel de confianza	0.9
4								Media propuesta	39
5								Tamaño de muestra	6
6								Media muestral	39.0113
7								Desviación estándar muestral	0.0119
8									
9								t0	2.3275
10								t crítica	2.0150
11									RECHAZAR H0
12									
13								p	0.0674
14								p crítica	0.1000
15									RECHAZAR H0

Celda	Fórmula	Resultado
I13	=T.DIST.2T(I9,I5-1)	0.0674
I14	=1-I3	0.1000
I15	=IF(I13>I14,"NO RECHAZAR H0","RECHAZAR H0")	RECHAZAR H0

En este caso, se rechazará H_0 si el valor crítico de p es menor al nivel de significancia deseado, el cual equivale a $1 - 0.9 = 0.1$.

Observe cómo los dos enfoques dan el mismo resultado: se debe rechazar H_0 por lo que se concluye que la media del proceso es diferente de 39 mm y la máquina requiere calibración.

=====

EL LIBRO COMPLETO INCLUYE MUCHOS MÁS EJERCICIOS DE ESTADÍSTICA EN EXCEL Y LAS HOJAS DE CÁLCULO UTILIZADAS PARA RESOLVERLOS.

[DESCARGAR OTRA VISTA PREVIA](#)

[DESCARGAR LIBRO COMPLETO](#)